
	Reason for change:
	The current TTCN implementation doesn’t take into account that a UE may use different TIs for call related and non-call related procedures. This needs to be accommodated. Also, In the current TTCN implementation, the AT command prompts the user to initiate a registration of call forwarding service for CFNRy(speech) at Step 1. In this, the noReplyConditionTimer is set as 15. The TTCN expects the UE to transmit a REGISTER message (Step 4) with the noReplyconditionTimer set as 15. However, as per 24.080 this IE is an optional IE. Some UEs may not send it. The TTCN currently expects that this IE will always be sent by the UE. This needs to be corrected. Similar change needs to be done for Step 9 when the AT command prompts the user to initiate a registration of call forwarding service for CFU (Speech) at Step 9 but does NOT include the noReplyConditionTimer. The REGISTER message at Step 12 sets the IE noReplyConditionTimer to OMIT, which needs to be corrected to AnyValue

	
	

	Summary of change:
	Created four new variables v_TI_R_CC, v_TI_S_CC, v_TI_R_SS and v_TI_S_SS to handle the possibility that a UE may use different TIs for call related and non-call related procedures. In the implementation to receive the REGISTER message at Step 4, replaced v_NoReplyConditionTime with *. In the implementation of REGISTER message at Step 12, in the template cr_InvokeCFU passed * as argument for  noReplyconditionTimer
Please see screenshots.

2 Correction to Supplimentary Service testcase 15.4.2
2.1 f_TC_15_4_2_UTRAN() 
	Object name 
	f_TC_15_4_2_UTRAN()

	Reason for change
	The current TTCN implementation doesn’t take into account that a UE may use different TIs for call related and non-call related procedures. This needs to be accommodated.

	Summary of change
	Created four new variables v_TI_R_CC, v_TI_S_CC, v_TI_R_SS and v_TI_S_SS to handle the possibility that a UE may use different TIs for call related and non-call related procedures.
Please see screenshots

	Module
	CallForwarding.ttcn

	Label
	WA#


Before:
	function f_TC_15_4_2_UTRAN() runs on UTRAN_PTC
  { /* Call forwarding supplementary services, Registration rejected */
    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;
    var octetstring v_EncodedComponents;
    var charstring v_ForwardedToNumber:= "01234556789"; //Any number
    var octetstring v_ForwardedToNumber_ForASN1 := tsc_NumberingPlan_Oct & f_CharPacking_IA5toBCD(v_ForwardedToNumber); // @sic R5s150538 sic@
    var NoReplyConditionTime v_NoReplyConditionTime := 15; //Any number
    var charstring v_NoReplyConditionTime_Charstring := int2char(v_NoReplyConditionTime);
    var Components v_FacilityInvoke;
    var octetstring v_FacilityReturn;
    var TI v_TI_R := f_NAS_TI_Init(cs_TI_MT);
    var TI v_TI_S;
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    f_UTRAN34_Preamble(utran34_Cell1);
    // The UE is in call state U10 by using table 10.1.2/3
    v_TI_R := f_UTRAN34_GoToState6_9_MO(utran34_Cell1, true); //Establishment of an outgoing call TS34.123, Table 10.1.2/3
    f_UTRAN_TestBody_Set(true);
    //If UE supports CFNRy, then step 1 to 8 are executed
    if (pc_SS_CallForwardinoNoReply) {
      // @siclog "Step 1" siclog@
      // The UE is made to initiate a registration of call forwarding service for CFNRy (Speech)"
      f_UT_SupplementaryService_CallForward(UT,
                                            REGISTER_MODE,
                                            SS_CFNRy,
                                            v_ForwardedToNumber,
                                            v_NoReplyConditionTime_Charstring,
                                            SS_SPEECH);
      // @siclog "Steps 2-4" siclog@
      // CM SERVICE REQUEST - CM SERVICE ACCEPT - REGISTER
      v_ReceiveRegisterResult := f_SS_ReceiveRegister_UE_Active(cr_InvokeCFNRy(v_ForwardedToNumber_ForASN1, v_NoReplyConditionTime), // @sic R5s150538 sic@
                                                                "Step 2",
                                                                "Step 4");
      v_FacilityInvoke := v_ReceiveRegisterResult.components;
      v_TI_R := v_ReceiveRegisterResult.ti;
      // Set TI value for response to be sent from the SS
      v_TI_S := f_NAS_TI_InitAndSetFlag(v_TI_R, '1'B);
      // Assign invokeID in the ReturnError with the value received in the Invoke
      v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnError(cs_ReturnErrorCode (v_FacilityInvoke[0].invoke.invokeID, tsc_ErrorCode_BearerServiceNotProvisioned))));
      // @siclog "Step 5" siclog@
      // SS sends RELEASE COMPLETE with Return error component with BearerServiceNotProvisioned
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_ReleaseCompleteFacility(v_TI_S, cs_FacilityIE_tlv(v_FacilityReturn))));
      // @siclog "Step 6" siclog@
      // Provide user indication
      if (f_UT_ChkSuppServiceUserIndication(UT, "CFNRy", "Registration failure due to BearerService Not Provisioned")) {
        f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");
      } else {
        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
      }
      // @siclog "Step 7-8" siclog@
      f_UTRAN34_CheckStatus(v_TI_S, v_TI_R, "Step 8", cr_CauResponseToStatusEnquiry, tsc_CallState_U10);
    }
    //If UE supports CFU, then step 9 to 16 are executed
    if (pc_SS_CallForwardingUnconditional) {
      // @siclog "Step 9" siclog@
      // The UE is made to initiate a registration of call forwarding service for CFU (Speech)
      f_UT_SupplementaryService_CallForward(UT,
                                            REGISTER_MODE,
                                            SS_CFU,
                                            v_ForwardedToNumber,
                                            -,
                                            SS_SPEECH);
      // @siclog "Steps 10-12" siclog@
      // CM SERVICE REQUEST - CM SERVICE ACCEPT - REGISTER
      v_ReceiveRegisterResult := f_SS_ReceiveRegister_UE_Active(cr_InvokeCFU(v_ForwardedToNumber_ForASN1), // @sic R5s150538 sic@
                                                                "Step 10",
                                                                "Step 12");
      v_FacilityInvoke := v_ReceiveRegisterResult.components;
      v_TI_R := v_ReceiveRegisterResult.ti;
      // Set TI value for response to be sent from the SS
      v_TI_S := f_NAS_TI_InitAndSetFlag(v_TI_R, '1'B);
      // @siclog "Step 13" siclog@
      // RegisterSS operation Return reject component with cause "resource limitation"
      v_EncodedComponents := f_SS_EncodeComponents(cs_Components(cs_ReturnReject(cs_Reject_ResourceLimitation(v_FacilityInvoke[0].invoke.invokeID))));
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_ReleaseCompleteFacility(v_TI_S, cs_FacilityIE_tlv(v_EncodedComponents)))); // @sic R5s150540 sic@
      // @siclog "Step 14" siclog@
      // Provide user indication
      if (f_UT_ChkSuppServiceUserIndication(UT, "CFU", "Registration failure due to Resource limitation")) {
        f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 14");
      } else {
        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
      }
      // @siclog "Step 15-16" siclog@
      f_UTRAN34_CheckStatus(v_TI_S, v_TI_R, "Step 16", cr_CauResponseToStatusEnquiry, tsc_CallState_U10);
    }
    //Postamble
    f_UTRAN_TestBody_Set(false);
    f_UTRAN_CC_Disconnect(utran34_Cell1, v_TI_S);
    f_UTRAN34_SwitchPower_Off(utran34_Cell1, U2_CONNECTED, NotSpecified);
    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType(utran34_Cell1));
  }
	


After:
	function f_TC_15_4_2_UTRAN() runs on UTRAN_PTC
  { /* Call forwarding supplementary services, Registration rejected */
    var SS_ReceiveRegisterResult_Type v_ReceiveRegisterResult;
    var octetstring v_EncodedComponents;
    var charstring v_ForwardedToNumber:= "01234556789"; //Any number
    var octetstring v_ForwardedToNumber_ForASN1 := tsc_NumberingPlan_Oct & f_CharPacking_IA5toBCD(v_ForwardedToNumber); // @sic R5s150538 sic@
    var NoReplyConditionTime v_NoReplyConditionTime := 15; //Any number
    var charstring v_NoReplyConditionTime_Charstring := int2char(v_NoReplyConditionTime);
    var Components v_FacilityInvoke;
    var octetstring v_FacilityReturn;
    var TI v_TI_R := f_NAS_TI_Init(cs_TI_MT);
    var TI v_TI_S;    
    //WA#15_4_2: introduced additional variables to consider the possibility that the UE may use different TI values for call related and non call related supplementary services
    var TI v_TI_R_CC; 
    var TI v_TI_S_CC;
    var TI v_TI_R_SS;
    var TI v_TI_S_SS;
    //WA#15_4_2 : Stop
    //Initialise all UTRAN cells according to 34.108
    f_UTRAN34_Init();
    f_UTRAN34_SS_CreateCellDCH(utran34_Cell1);
    f_UTRAN34_SendDefSysInfo(utran34_Cell1);
    f_UTRAN34_Preamble(utran34_Cell1);
    // The UE is in call state U10 by using table 10.1.2/3
    v_TI_R_CC /*WA#15_4_2 v_TI_R*/ := f_UTRAN34_GoToState6_9_MO(utran34_Cell1, true); //Establishment of an outgoing call TS34.123, Table 10.1.2/3
    f_UTRAN_TestBody_Set(true);
    //If UE supports CFNRy, then step 1 to 8 are executed
    if (pc_SS_CallForwardinoNoReply) {
      // @siclog "Step 1" siclog@
      // The UE is made to initiate a registration of call forwarding service for CFNRy (Speech)"
      f_UT_SupplementaryService_CallForward(UT,
                                            REGISTER_MODE,
                                            SS_CFNRy,
                                            v_ForwardedToNumber,
                                            v_NoReplyConditionTime_Charstring,
                                            SS_SPEECH);
      // @siclog "Steps 2-4" siclog@
      // CM SERVICE REQUEST - CM SERVICE ACCEPT - REGISTER
      v_ReceiveRegisterResult := f_SS_ReceiveRegister_UE_Active(cr_InvokeCFNRy(v_ForwardedToNumber_ForASN1, 
                                                                               * /*WA#15_4_2 v_NoReplyConditionTime*/), // @sic R5s150538 sic@
                                                                "Step 2",
                                                                "Step 4");
      v_FacilityInvoke := v_ReceiveRegisterResult.components;
      v_TI_R_SS /*WA#15_4_2 v_TI_R*/ := v_ReceiveRegisterResult.ti;
      // Set TI value for response to be sent from the SS
      v_TI_S_SS /*WA#15_4_2 v_TI_S*/ := f_NAS_TI_InitAndSetFlag(v_TI_R_SS /*WA#15_4_2 v_TI_R*/, '1'B);
      v_TI_S_CC := f_NAS_TI_InitAndSetFlag(v_TI_R_CC, '1'B);//WA#15_4_2
      // Assign invokeID in the ReturnError with the value received in the Invoke
      v_FacilityReturn := f_SS_EncodeComponents(cs_Components(cs_ReturnError(cs_ReturnErrorCode (v_FacilityInvoke[0].invoke.invokeID, tsc_ErrorCode_BearerServiceNotProvisioned))));
      // @siclog "Step 5" siclog@
      // SS sends RELEASE COMPLETE with Return error component with BearerServiceNotProvisioned
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_ReleaseCompleteFacility(v_TI_S_SS/*WA#15_4_2 v_TI_S*/, cs_FacilityIE_tlv(v_FacilityReturn))));
      // @siclog "Step 6" siclog@
      // Provide user indication
      if (f_UT_ChkSuppServiceUserIndication(UT, "CFNRy", "Registration failure due to BearerService Not Provisioned")) {
        f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");
      } else {
        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 6");
      }
      // @siclog "Step 7-8" siclog@
      f_UTRAN34_CheckStatus(v_TI_S_CC/*WA#15_4_2 v_TI_S*/, v_TI_R_CC/*WA#15_4_2 v_TI_R*/, "Step 8", cr_CauResponseToStatusEnquiry, tsc_CallState_U10);
    }
    //If UE supports CFU, then step 9 to 16 are executed
    if (pc_SS_CallForwardingUnconditional) {
      // @siclog "Step 9" siclog@
      // The UE is made to initiate a registration of call forwarding service for CFU (Speech)
      f_UT_SupplementaryService_CallForward(UT,
                                            REGISTER_MODE,
                                            SS_CFU,
                                            v_ForwardedToNumber,
                                            -,
                                            SS_SPEECH);
      // @siclog "Steps 10-12" siclog@
      // CM SERVICE REQUEST - CM SERVICE ACCEPT - REGISTER
      v_ReceiveRegisterResult := f_SS_ReceiveRegister_UE_Active(cr_InvokeCFU(v_ForwardedToNumber_ForASN1, * /*WA#15_4_2*/), // @sic R5s150538 sic@
                                                                "Step 10",
                                                                "Step 12");
      v_FacilityInvoke := v_ReceiveRegisterResult.components;
      v_TI_R_SS/*WA#15_4_2 v_TI_R*/ := v_ReceiveRegisterResult.ti;
      // Set TI value for response to be sent from the SS
      v_TI_S_SS/*WA#15_4_2 v_TI_S*/ := f_NAS_TI_InitAndSetFlag(v_TI_R_SS/*WA#15_4_2 v_TI_R*/, '1'B);
      v_TI_S_CC := f_NAS_TI_InitAndSetFlag(v_TI_R_CC, '1'B);//WA#15_4_2
      // @siclog "Step 13" siclog@
      // RegisterSS operation Return reject component with cause "resource limitation"
      v_EncodedComponents := f_SS_EncodeComponents(cs_Components(cs_ReturnReject(cs_Reject_ResourceLimitation(v_FacilityInvoke[0].invoke.invokeID))));      
      U_Dc.send(cas_DataReq_CS(utran_CellDedicated,
                               tsc_RB3,
                               cs_U_ReleaseCompleteFacility(v_TI_S_SS/*WA#15_4_2 v_TI_S*/, cs_FacilityIE_tlv(v_EncodedComponents)))); // @sic R5s150540 sic@  
      // @siclog "Step 14" siclog@
      // Provide user indication
      if (f_UT_ChkSuppServiceUserIndication(UT, "CFU", "Registration failure due to Resource limitation")) {
        f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 14");
      } else {
        f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 14");
      }
      // @siclog "Step 15-16" siclog@
      f_UTRAN34_CheckStatus(v_TI_S_CC/*WA#15_4_2 v_TI_S*/, v_TI_R_CC/*WA#15_4_2 v_TI_R*/, "Step 16", cr_CauResponseToStatusEnquiry, tsc_CallState_U10);
    }
    //Postamble
    f_UTRAN_TestBody_Set(false);
    f_UTRAN_CC_Disconnect(utran34_Cell1, v_TI_S_CC/*WA#15_4_2 v_TI_S*/);
    f_UTRAN34_SwitchPower_Off(utran34_Cell1, U2_CONNECTED, NotSpecified);
    f_UTRAN_ReleaseCell(utran34_Cell1, f_UTRAN_CellInfo_GetConfigType(utran34_Cell1));
  }


	

	


